demandsys postestimation — Postestimation tools for demandsys

Postestimation commands

The following postestimation commands are of special interest after demandsys:

<table>
<thead>
<tr>
<th>Command</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>estat elasticities</td>
<td>price and expenditure elasticities</td>
</tr>
<tr>
<td>estat parameters</td>
<td>estimated parameter vectors and matrices</td>
</tr>
</tbody>
</table>

The following standard postestimation commands are also available:

<table>
<thead>
<tr>
<th>Command</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>estat ic</td>
<td>Akaike’s, consistent Akaike’s, corrected Akaike’s, and Schwarz’s Bayesian</td>
</tr>
<tr>
<td></td>
<td>information criteria (AIC, CAIC, AICc, and BIC)</td>
</tr>
<tr>
<td>estat summarize</td>
<td>summary statistics for the estimation sample</td>
</tr>
<tr>
<td>estat vce</td>
<td>variance–covariance matrix of the estimators (VCE)</td>
</tr>
<tr>
<td>estimates</td>
<td>cataloging estimation results</td>
</tr>
<tr>
<td>lincom</td>
<td>point estimates, standard errors, testing, and inference for linear</td>
</tr>
<tr>
<td></td>
<td>combinations of coefficients</td>
</tr>
<tr>
<td>lrtest</td>
<td>likelihood-ratio test</td>
</tr>
<tr>
<td>margins</td>
<td>marginal means, predictive margins, marginal effects, and average</td>
</tr>
<tr>
<td></td>
<td>marginal effects</td>
</tr>
<tr>
<td>marginsplot</td>
<td>graph the results from margins (profile plots, interaction plots, etc.)</td>
</tr>
<tr>
<td>nlcom</td>
<td>point estimates, standard errors, testing, and inference for nonlinear</td>
</tr>
<tr>
<td></td>
<td>combinations of coefficients</td>
</tr>
<tr>
<td>predict</td>
<td>predicted shares, residuals, quantities, and functions</td>
</tr>
<tr>
<td>predictnl</td>
<td>point estimates, standard errors, testing, and inference for generalized</td>
</tr>
<tr>
<td></td>
<td>predictions</td>
</tr>
<tr>
<td>test</td>
<td>Wald tests of simple and composite linear hypotheses</td>
</tr>
<tr>
<td>testnl</td>
<td>Wald tests of nonlinear hypotheses</td>
</tr>
</tbody>
</table>
predict

Description for predict
predict creates a new variable containing statistics such as predicted shares and residuals based on the demandsys estimation results currently active.

Menu for predict
Statistics > Postestimation

Syntax for predict
predict [type] newvar [if] [in] [, statistic options]

<table>
<thead>
<tr>
<th>statistic</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>shares</td>
<td>predicted shares; the default</td>
</tr>
<tr>
<td>quantities</td>
<td>predicted quantities</td>
</tr>
<tr>
<td>residuals</td>
<td>residuals</td>
</tr>
<tr>
<td>iuf</td>
<td>indirect utility function</td>
</tr>
<tr>
<td>ef</td>
<td>expenditure function</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>options</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>equation(#eqno)</td>
<td>display only specified goods; available only with shares, quantities, and residuals</td>
</tr>
<tr>
<td>prices(varlist_p)</td>
<td>specify set of counterfactual price variables</td>
</tr>
<tr>
<td>expenditures(varname)</td>
<td>specify a counterfactual total expenditure variable</td>
</tr>
<tr>
<td>utilities(varname)</td>
<td>specify the variable containing the utilities at which to compute the expenditure function; available only with and required with ef</td>
</tr>
<tr>
<td>demographics(varlist)</td>
<td>specify a set of counterfactual demographic variables</td>
</tr>
</tbody>
</table>

Options for predict
shares, the default, calculates the predicted shares.
quantities calculates the predicted quantities given actual prices, actual expenditures, and predicted shares.
residuals calculates the residuals.
iuf calculates the indirect utility function $v(p, m)$.
ef calculates the expenditure function $e(p, u)$. You must also specify option utilities() with this option. This option is not available after translog and generalized translog models.
equation(#eqno) specifies to which good you are referring. equation(#1) would mean the calculation is to be made for the first good, equation(#2) would mean the second, and so on. If you do not specify equation(), results are the same as if you specified equation(#1).

prices(varlist) specifies a set of counterfactual price variables at which to compute the requested statistic. You must specify $G$ variables, where $G$ is the number of goods in the demand system. By default, predict uses the same price variables that were used to fit the demand system. This option aids in the computation of welfare and other policy measures by allowing you to contrast predicted values at different price levels.

expenditures(varname) specifies a counterfactual total expenditure variable at which to compute the requested statistic. By default, predict uses the same expenditure variable that was used to fit the demand system. This option aids in the computation of welfare and other policy measures by allowing you to contrast predicted values at different expenditure levels.

utilities(varname) specifies the variable containing the utilities at which the expenditure function is to be computed. Almost surely, varname is a variable that was created by calling predict..., iuf... This option aids in the computation of welfare and other policy measures by allowing you to contrast predicted values at different levels of utility.

demographics(varlist) specifies a set of counterfactual demographic variables at which to compute the requested statistic. If you specified $D$ demographic variables at estimation, then you must specify precisely $D$ variables here. By default, predict uses the same demographic variables (if any) that were used to fit the demand system.

margins

Description for margins

margins estimates margins for the predicted share of the first good.

Menu for margins

Statistics > Postestimation

Syntax for margins

margins [, options]

margins, predict(statistic ...) [options]

<table>
<thead>
<tr>
<th>statistic</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>shares</td>
<td>predicted shares; the default</td>
</tr>
<tr>
<td>quantities</td>
<td>predicted quantities</td>
</tr>
<tr>
<td>residuals</td>
<td>residuals</td>
</tr>
<tr>
<td>iuf</td>
<td>indirect utility function</td>
</tr>
<tr>
<td>*ef</td>
<td>expenditure function</td>
</tr>
</tbody>
</table>

*You must specify the predict option utilities() with statistic ef.

Statistics not allowed with margins are functions of stochastic quantities other than e(b).

For the full syntax, see [R] margins.
estat

Description for estat

estat elasticities computes expenditure and price elasticities after fitting a demand system using demandsys. Elasticities are calculated at the price and expenditure levels in the data. However, you may compute elasticities at the means of prices and expenditures or use other counterfactual prices or expenditures.

estat parameters returns the estimated parameter vectors and matrices based on the currently active demandsys estimation results. Results are stored in \( r() \) and vary depending on the demand model fitted. See [U] 18.8 Accessing results calculated by other programs.

Menu for estat

Statistics > Postestimation

Syntax for estat

Compute price and expenditure elasticities

\[
\text{estat elasticities } [\text{if}] \ [\text{in}], \ {\text{expenditure|compensated|uncompensated}} \\
\quad \quad [[\text{atmeans}|\text{at}(\text{atspec})] \ \text{generate}(\text{stub})]
\]

Report estimated parameter vectors and matrices

estat parameters

Options for estat elasticities

Main

expenditure, compensated, and uncompensated indicate the type of elasticities to be reported. You must specify one of these.

expenditure requests that the \( G \) expenditure elasticities be calculated, where \( G \) is the number of goods in the system.

compensated requests that compensated price elasticities be calculated. There is a total of \( G^2 \) such elasticities because there are \( G \) goods, each of whose quantity consumed can be affected by any one of the \( G \) prices changing.

uncompensated requests that the \( G^2 \) uncompensated price elasticities be calculated.

atmeans requests that elasticities be calculated at the means of the price, expenditure, and demographic variables. By default, elasticities are computed for each observation and then averaged.

at(atspec) specifies the prices, expenditures, and demographic variables’ values at which elasticities are to be calculated. If you do not specify this option, elasticities are calculated at the observed values unless you specify option atmeans, in which case they are calculated at the means of the variables. See example 1 below. You may not specify option at() if you request compensated price elasticities.
generate(stub) requests that new variables be generated holding the observation-level elasticities. For expenditure elasticities, $G$ new variables $stub_1, stub_2, \ldots, stub_G$ will be created, where $G$ is the number of goods. For price elasticities, $G^2$ new variables will be created of the form $stub_g_h$ for $g = 1, \ldots, G$ and $h = 1, \ldots, G$. Variable $stub_g_h$ will contain the percentage change in the quantity of good $h$ with respect to the price of good $g$.

Remarks and examples

Remarks are presented under the following headings:

- Introduction
- Elasticities
- Evaluating elasticities
- Compensating and equivalent variation

Introduction

`estat elasticities` allows you to compute expenditure and price elasticities after you fit a demand model and is much more flexible than the options available with `demandsys` to display elasticities in lieu of coefficient estimates. With `estat elasticities`, you can calculate elasticities at the same prices, expenditures, and demographic variables’ levels as was used at estimation time, or you can specify your own values. For example, you could calculate price elasticities for every observation in your dataset assuming each household had two children, even if the number of children varies by household. Example 1 below shows you how that can be done.

`predict` allows you to obtain values of the indirect utility function and of the expenditure function underlying the demand system you fit. Example 3 leverages that ability and shows you how to compute compensated and equivalent variation in response to a price change.

`estat parameters` allows you to obtain the estimated parameter vectors and matrices for the demand system based on the coefficient vector stored in `e(b)`. This may be useful if you wish to conduct further policy experiments or calculate your own welfare measures. For instance, in example 2 below, we show how `estat parameters` with just a bit of coding can be used to check whether a fitted generalized translog model can be interpreted as having subsistence quantities for each good.

Elasticities

Although one can have `demandsys` report expenditure or price elasticities for the estimation sample, the `estat elasticities` command gives you much more flexibility. We illustrate features of `estat elasticities` in the next example. We often request expenditure elasticities to keep the output shorter, but all our commands would work regardless of the type of elasticity requested.

Example 1

Before we can obtain elasticities, we first need to fit a demand system to our data. Here we include the `quietly` prefix command to omit the results of fitting a quadratic almost ideal demand system (QUAIDS) model with demographic translation, and we use the `labels()` option to provide short names for each of our goods. We type
. use https://www.stata-press.com/data/r18/food_consumption
(\text{Food consumption})
. quietly demandsys qualds w_dairy w_proteins w_fruitveg w_flours w_misc,
> \text{prices(p_dairy p_proteins p_fruitveg p_flours p_misc) }
> \text{expenditures(expfd)}
> \text{demographics(n_kids n_adults)}
> \text{labels(Dairy Meats FruitVeg Flours Misc)}

Suppose we want the average expenditure elasticity among households with two children. We type
\text{. estat elasticities if n\_kids == 2, expenditure}

\begin{table}[h]
\centering
\begin{tabular}{lcccrr}
\hline
Expenditure & Elasticity & Std. err. & \text{z} & P>|z| & \text{[95\% conf. interval]} \\
\hline
\text{Dairy} & \text{0.9028921} & \text{0.0240869} & \text{37.48} & \text{0.000} & \text{[0.8556826, 0.9501016]} \\
\text{Meats} & \text{1.087763} & \text{0.0156852} & \text{69.35} & \text{0.000} & \text{[1.05702, 1.118505]} \\
\text{FruitVeg} & \text{0.9978958} & \text{0.0215996} & \text{46.20} & \text{0.000} & \text{[0.9555614, 1.04023]} \\
\text{Flours} & \text{0.8523383} & \text{0.0220511} & \text{38.65} & \text{0.000} & \text{[0.8091189, 0.8955576]} \\
\text{Misc} & \text{0.9971274} & \text{0.0265993} & \text{37.49} & \text{0.000} & \text{[0.9449938, 1.049261]} \\
\hline
\end{tabular}
\end{table}

The header of the output reports that there are 751 households in the dataset with two children. Had
we not specified the \text{labels()} option when we fit our model, \text{estat elasticities} would have
simply numbered the goods from one through five.

What is displayed in the table of output is an average elasticity. Behind the scenes, \text{estat elasticities}
calculated the five expenditure elasticities for each of the 751 households in this
sample. What is reported as the expenditure elasticity for dairy products is the average over the 751
individual expenditure elasticities calculated. The column labeled \text{Std. err.} contains the standard
error of the estimated mean. Looking to the right, we see the 95\% confidence interval for the mean
dairy elasticity is [0.856, 0.950]. More importantly, the column labeled \text{Std. err.} is not the sample
standard deviation of the 751 individual elasticities, nor does the 95\% confidence interval report
something resembling the central 95\% of the individual elasticities.

To save the individual elasticities for the 751 observations, you can use the \text{generate()} option
like so:
\text{. quietly estat elasticities if n\_kids == 2, expenditure generate(ee)}
\text{. summarize ee*}

\begin{tabular}{lrrrr}
\hline
Variable & Obs & Mean & Std. dev. & Min & Max \\
\hline
\text{ee\_1} & 751 & \text{0.9028921} & \text{0.0753773} & \text{0.5095888} & \text{1.44884} \\
\text{ee\_2} & 751 & \text{1.087763} & \text{0.0664746} & \text{0.8448506} & \text{1.525114} \\
\text{ee\_3} & 751 & \text{0.9978958} & \text{0.1755427} & \text{0.7566877} & \text{5.07766} \\
\text{ee\_4} & 751 & \text{0.8523383} & \text{0.1053903} & \text{0.3864381} & \text{1.110392} \\
\text{ee\_5} & 751 & \text{0.9971274} & \text{0.082875} & \text{0.3257591} & \text{1.296199} \\
\hline
\end{tabular}

The means of these variables match the point estimates in the previous \text{estat elasticities} output.
The standard deviations reported here measure the spread of the individual observations’ elasticities
rather than the standard errors of the means as in the previous output.

A second way to compute elasticities is to first compute the means of the price variables, the
expenditure variable, and any demographics and then compute the elasticity for this hypothetical
household. To have \text{estat elasticities} do that, you use the \text{atmeans} option. Here we obtain the
expenditure elasticities at the means for the subset of households with two children:
demandsys postestimation — Postestimation tools for demandsys

```
. estat elasticities if n_kids == 2, expenditure atmeans
Expenditure elasticities

| Expenditure | Elasticity  | Std. err. | z    | P>|z| | [95% conf. interval] |
|-------------|-------------|-----------|------|-------|----------------------|
| Dairy       | .9155493    | .0232877  | 39.31| 0.000 | .8699063             | .9611923 |
| Meats       | 1.078329    | .0149179  | 72.28| 0.000 | 1.04909              | 1.107567 |
| FruitVeg    | .9804481    | .0201581  | 48.64| 0.000 | .9409389             | 1.019957 |
| Flours      | .8695964    | .0212882  | 40.85| 0.000 | .8278722             | .9113206 |
| Misc        | 1.004098    | .0246939  | 40.66| 0.000 | .9556994             | 1.052498 |
```

Note: Elasticities are calculated at prices’, demographic variables’, and expenditure means.

With the `atmeans` option, we calculate the elasticities for a hypothetical household that may have, for example, 0.87 children and 1.93 adults. Without the `atmeans` option, we are calculating elasticities for households that do exist in the data, though taking the average of household elasticities poses its own conceptual issues.

Consider the following table:

<table>
<thead>
<tr>
<th>Consumer</th>
<th>$m$</th>
<th>$E^m$</th>
<th>$Q$</th>
<th>$m'$</th>
<th>$Q'$</th>
</tr>
</thead>
<tbody>
<tr>
<td>A</td>
<td>100</td>
<td>2</td>
<td>20</td>
<td>110</td>
<td>24</td>
</tr>
<tr>
<td>B</td>
<td>200</td>
<td>1</td>
<td>30</td>
<td>220</td>
<td>33</td>
</tr>
<tr>
<td>C</td>
<td>400</td>
<td>0.5</td>
<td>30</td>
<td>440</td>
<td>31.5</td>
</tr>
<tr>
<td>Sum</td>
<td>700</td>
<td>80</td>
<td>770</td>
<td>88.5</td>
<td></td>
</tr>
</tbody>
</table>

Mean: 1.17

Column $m$ represents each consumer’s initial total expenditure, $E^m$ is the consumer’s expenditure elasticity for the good in question, and $Q$ is the quantity consumed. Column $m'$ represents a 10% increase in each consumer’s expenditure, and $Q'$ represents the level of consumption based on the increases in total expenditure and the expenditure elasticities. The mean elasticity, analogous to what `estat elasticities` reports without the `atmeans` option, is 1.17. However, based on the aggregate data, the expenditure elasticity works out to 1.06.

So far in this example, we computed the expenditure elasticities for the subset of households with two children. We can also calculate the average elasticity for all households in our dataset under the counterfactual that all of them have two children. One way to do that is to type

```
. generate n_kids_backup = n_kids
. replace n_kids = 2
   (3,409 real changes made)
. estat elasticities, expenditure
```

```
Expenditure elasticities

| Expenditure | Elasticity  | Std. err. | z    | P>|z| | [95% conf. interval] |
|-------------|-------------|-----------|------|-------|----------------------|
| Dairy       | .8464408    | .0266003  | 31.82| 0.000 | .7943051             | .8985764 |
| Meats       | 1.20188     | .0252998  | 47.51| 0.000 | 1.152294             | 1.251467 |
| FruitVeg    | 1.130297    | .0323216  | 34.97| 0.000 | 1.066948             | 1.193646 |
| Flours      | .7604338    | .0229378  | 33.15| 0.000 | .7154766             | .8053911 |
| Misc        | .9082412    | .0287833  | 31.55| 0.000 | .8518269             | .9646554 |
```

...
Here we first created a backup variable holding the true values of `n_kids`, and then we set `n_kids` equal to two for all observations. After calling `estat elasticities`, we restored the true values.

**Technical note**

In principle, a second way to obtain the expenditure elasticities under the counterfactual that all households have two children would be to type

```
estat elasticities, expenditure at(n_kids=2)
```

The key is the `at()` option, which allows us to request that variables be set to the values we specify before calculating the elasticities. For more information about the `at()` option, see [R] `margins`, especially *Syntax of at().*

However, in this example, if you issue the previous command, you would obtain the error message

```
. estat elasticities, expenditure at(n_kids=2)
Option at() specified; using margins.
could not calculate numerical derivatives -- discontinuous region with missing
values encountered
r(459);
```

The issue is that for most demand systems the predicted quantities are nonlinear functions of prices. As the output indicates, when you specify the `at()` option, `estat elasticities` calls `margins` to obtain the desired results. Depending on the exact demand system and demographic specification, `margins` may not be able to obtain all the required derivatives numerically.

When you do not use the `at()` option, `estat elasticities` uses analytics formulas for the expenditure and price elasticities; *Methods and formulas* of [R] `demandsys` includes the elasticity formulas for all the demand systems implemented. Evaluating analytic formulas is quick, and most of the time is spent computing the variance–covariance matrix of the estimated mean elasticities, because that does use numerical derivatives. Because those derivatives are for a sample mean, rather than individual observations, you are unlikely to obtain the same message as `margins` issued in the previous command.

In contrast, `margins` uses numerical derivatives for all its calculations, and, moreover, it must incur the added expense of having to call `predict` to evaluate predicted quantities. One case where you would need to use `margins` is to obtain, say, a quantity elasticity with respect to a change in a demographic variable, because those elasticities are not programmed into `demandsys`.

In the text after example 4 and in example 6 of [R] `demandsys`, we cautioned that after fitting a generalized model that allows for subsistence or committed quantities, you should check whether that interpretation actually holds. If for some household \( \sum_g p_g (\mu_g + \nu'_g d) > m \), then either the household is not consuming enough to meet what we think is a minimum subsistence or our assumption that this quantity represents a minimum subsistence is wrong. Moreover, if \( \mu_g + \nu'_g d < 0 \), then we are saying that the minimum required level of \( g \) is less than 0, which implies that good \( g \) is in fact not needed for subsistence.
Example 2

This example is a continuation of example 6 in [R] demandsys. In that example, we fit a generalized QUAIDS model, which we refit here:

```
. use https://www.stata-press.com/data/r18/food_consumption, clear
   (Food consumption)
. demandsys gquaids w_dairy w_proteins w_fruitveg w_flours w_misc,
   > prices(p_dairy p_proteins p_fruitveg p_flours p_misc)
   > demographics(n_kids n_adults) pconstant(3) expenditure(expfd) nolog
   (output omitted)
```

Next, we compute variables containing each household’s implied minimum quantity of each good conditional on that household’s demographics. Given those variables, we can then calculate the expenditure needed to acquire the minimum quantities given the prices each household faces. We type

```
. estat parameters
. return list
macros:
   r(names) : "alpha beta Gamma lambda mu Nu"
matrices:
   r(Nu) : 5 x 2
   r(mu) : 5 x 1
   r(lambda) : 5 x 1
   r(Gamma) : 5 x 5
   r(beta) : 5 x 1
   r(alpha) : 5 x 1

. matrix mu = r(mu)
. matrix Nu = r(Nu)
. forvalues g = 1/5 {
   2. generate minq_`g' = mu[`g', 1] + Nu[`g',1]*n_kids + Nu[`g',2]*n_adults
   3. }
. generate sub_exp = p_dairy*minq_1 + p_proteins*minq_2
   > + p_fruitveg*minq_3 + p_flour*minq_4 + p_misc*minq_5
```

We first issued the command `estat parameters` to obtain the estimated parameter matrices from our demand system. The following command shows that six vectors and matrices were stored, following the layout of the coefficient table from `demandsys`. To compute minimum quantities, we need matrices $\mu$ and $N$, so we transferred those $r()$ results to Stata matrices.

Because we specified `n_kids` first in the `demographics()` option of `demandsys`, that variable’s coefficients are stored in the first column of our matrix $r(Nu)$, which we converted to $Nu$. The second column corresponds to coefficients associated with `n_adults`. We looped over the five goods in our system and created variables named `minq_g` containing the minimum quantities. The final `generate` statement calculates the minimum expenditure for each household given the prices it faced and our calculated minimum quantities.
Now we can look at the results.

```
.summarize minq_*
```

<table>
<thead>
<tr>
<th>Variable</th>
<th>Obs</th>
<th>Mean</th>
<th>Std. dev.</th>
<th>Min</th>
<th>Max</th>
</tr>
</thead>
<tbody>
<tr>
<td>minq_1</td>
<td>4,160</td>
<td>-1.604524</td>
<td>.9955551</td>
<td>-7.287278</td>
<td>-.2078816</td>
</tr>
<tr>
<td>minq_2</td>
<td>4,160</td>
<td>-2.893996</td>
<td>1.743902</td>
<td>-13.71866</td>
<td>-.6731596</td>
</tr>
<tr>
<td>minq_3</td>
<td>4,160</td>
<td>-5.344703</td>
<td>3.802853</td>
<td>-29.30052</td>
<td>-.5518757</td>
</tr>
<tr>
<td>minq_4</td>
<td>4,160</td>
<td>.3094949</td>
<td>.1354879</td>
<td>-0.0171549</td>
<td>1.328885</td>
</tr>
<tr>
<td>minq_5</td>
<td>4,160</td>
<td>-.1878336</td>
<td>.2030023</td>
<td>-1.34631</td>
<td>.0894621</td>
</tr>
</tbody>
</table>

For the first three goods in our system, the calculated minimum quantity is negative for every single household in our sample. That is hardly encouraging if we wish to take the subsistence interpretation seriously, because it implies three of the five categories of food we have are not required for subsistence. Moreover, on average, good 5 is not required either. Even for good 4, for some households, the minimum quantity is negative. Finally, we look at the expenditure needed to purchase the minimum quantities of the five goods:

```
.summarize sub_exp
```

<table>
<thead>
<tr>
<th>Variable</th>
<th>Obs</th>
<th>Mean</th>
<th>Std. dev.</th>
<th>Min</th>
<th>Max</th>
</tr>
</thead>
<tbody>
<tr>
<td>sub_exp</td>
<td>4,160</td>
<td>-8.10118</td>
<td>5.480523</td>
<td>-43.8327</td>
<td>.2678578</td>
</tr>
</tbody>
</table>

Not surprisingly given the previous summary statistics, the minimum required expenditure for most households is less than zero.

What are we to make of the results in the previous example? We think there are at least four explanations:

1. There are minimum quantities of each good consumers must purchase for subsistence, but our demand system is misspecified. If you believe this to be true, you might try changing the demographic variables for which you control. Alternatively, try a different model; for example, use a generalized QUAIDS model rather than a generalized translog or AIDS model.

2. The data are simply not compatible with a minimum subsistence interpretation. That does not imply the results of the fitted demand system must be ignored. As Pollak and Wales (1992, 75) write, any demand system that includes demographic translation should include a constant term in the demand equations—which is the key characteristic of the models we refer to as “generalized”. For example, rather than using the translog model with demographic translation, you should instead use the generalized translog model with demographic translation. In other words, the $\mu$ parameters do belong in the model, but they should not be construed as minimum quantities required for subsistence.

3. Items for which the calculated minimum quantities are positive for all observations must be purchased to achieve subsistence, but items for which the minimum quantities are negative need not be consumed for subsistence. Under this interpretation, in our example, households must consume breads, cereals, and pastas but need not consume any other foods for minimal subsistence. That is not particularly palatable to us.

4. The procedure we used is useful as a diagnostic tool, but it is not a formal statistical test. Our estimated parameters are subject to sampling variation, and therefore our calculated minimum quantities are also subject to sampling variation. We have not accounted for this. If we were to conduct a test of the hypothesis that all minimum quantities are zero or more versus the alternative that some are negative, we may not be able to reject the null hypothesis.
Evaluating elasticities

The elasticities that are reported with `estat elasticities` refer to the slope of the fitted demand curve. Let us explain what we mean. Look at figure 1. We have plotted a fitted demand curve $D^F$ as a straight line for simplicity, and the large dots represent observed (price, quantity) pairs for two observations. The dotted lines represent each observation’s residual, and the plus signs represent the predicted quantities given the two observations’ prices. While quantity is a function of price here, following the influential English economist Alfred Marshall, we plot price on the ordinate; hence, the lines connecting observations to the regression line are horizontal, not vertical.

![Figure 1. Actual and predicted quantities](image)

For the AIDS model without demographic variables included, we can show that the expenditure elasticity for the $g$th good is given by

$$E^g = 1 + \frac{\beta_g}{w_g}$$

We do not observe the true value $\beta_g$, so to compute the elasticity for household $i$, we have no choice but to use the estimated $\hat{\beta}_g$. However, we observe both household $i$’s true expenditure share $w_{gi}$, and we can compute its predicted expenditure share $\hat{w}_{gi}$.

In terms of figure 1, the issue is whether we evaluate the expenditure or any other elasticity formula at the red and blue dots or at their corresponding fitted values shown by the pluses on the demand curve. We believe the correct procedure is to evaluate elasticity formulas at their fitted values. That is, `estat elasticities` evaluates the expenditure share elasticity for household $i$ as

$$\hat{E}^g_i = 1 + \frac{\hat{\beta}_g}{\hat{w}_{gi}}$$

After all, for policy or other analyses, what we want to know is essentially the slope of the fitted demand curve. We need to be able to predict the effect of a policy on a household, and in general to do that, we predict the household’s response with the policy imposed and compare it with the predicted response in the absence of the policy. In other words, our elasticity computations measure the slope of the fitted demand curve, not the slope of an imaginary line that passes through one of our dots that may or may not be parallel to the fitted demand curve.

In contrast, the postestimation tools accompanying the community-contributed `quaids` command of Poi (2012) use the actual values $w_g$ to calculate elasticities. Thus, users of that command will see...
minor differences when refitting models using `demandsys` and `estat elasticities`. We believe, however, that evaluating the formulas at the predicted expenditure shares, or equivalently, evaluating the elasticity formulas along the demand curve rather than at points in space that need not lie on the demand curve, is conceptually more sound. Moreover, the elasticity results reported after `quaisd` cannot be replicated using the `margins` command, but the results reported by `estat elasticities` are identical to those one would obtain using `margins`.

### Compensating and equivalent variation

Although students in introductory economics courses learn of consumer’s surplus as a way to measure whether consumers are better or worse off in response to a policy such as a tax, it has limitations. Perhaps most importantly, consumer’s surplus is not well defined in most cases; see Auerbach (1985). Also see Harberger (1964), who develops a method for calculating consumer welfare when multiple prices change but assumes the timing of the price changes is scattered rather than occurring simultaneously. Hines (1999) provides a historical overview of consumer’s welfare measurement.

The workhorse metrics used to evaluate a policy are compensating and equivalent variation (CV and EV, respectively). These metrics assign a monetary amount to a household’s change in welfare and can accommodate policies that cause multiple prices to change simultaneously. Because `predict` after `demandsys` allows us to obtain the predicted values of the expenditure and indirect utility functions, these metrics are easy to obtain.

CV measures how much a social planner would have to compensate a consumer to offset the effect of, say, a tax that causes prices to rise. Consider figure 2, which illustrates CV for two goods and would be familiar to those who have studied intermediate microeconomics.

![Figure 2. CV if the price of good 1 rises](image)

Initially, the price of good 1 is $p_0$, and the line segment $A \cdot p_0/m$ represents the budget constraint. The consumer purchases $q_0$ units of good 1 and enjoys utility level $U_0$. Now suppose the price of good 1 rises to $p_1$: consumption of good 1 falls to $q_1$ and utility falls to $U_1$; line segment $A \cdot p_1/m$ represents the new budget constraint.

The dotted line segment $B \cdot p_1'/m$ represents a budget constraint based on price $p_1$ but shifted outward so that the consumer can still achieve utility level $U_0$ when facing price $p_1$, by consuming $q_1'$. The line segment $B \cdot A$ represents the CV. It tells us, in the face of a higher price for good 1, how much the social planner would have to compensate this consumer so that he or she can enjoy
the same level of utility as before the price increase. In this case, \( CV \) is negative because the social planner must pay the consumer.

Notice the close similarity to the concept of the income and substitution effects that cause demand curves to slope downward. \( CV \), and \( EV \) as we will see shortly, is a measure of the income effect.

Mathematically, \( CV \) is defined as

\[
CV = e(p^1, u^1) - e(p^1, u^0) \\
= e(p^0, u^0) - e(p^1, u^0) \\
= m - e(p^1, u^0)
\]

where \( p^0 \) and \( p^1 \) are the price vectors before and after the policy is implemented and \( u^0 \) is the household's initial level of utility. Where do we get that level of utility, considering that utility is unobserved? The answer is the indirect utility function, which expresses utility as a function of observed prices and expenditure; we can evaluate the indirect utility function by calling \texttt{predict} with the \texttt{iuf} option. With the indirect utility variable in hand, we then call \texttt{predict} with the \texttt{ef} option and set the price variables to \( p^1 \).

\section*{Example 3}

We compute household \( CV \) in response to a 10\% increase in the price of dairy products, holding all other prices constant. See Poi (2002) for similar computations for an expanded look at the welfare implications of lower dairy and other prices. First, we fit a generalized QUAIDS model controlling for the number of children and adults in each household. We type

\begin{verbatim}
. use https://www.stata-press.com/data/r18/food_consumption, clear
(Food consumption)
. quietly demandsys gquaids w_dairy w_proteins w_fruitveg w_flours w_misc,
> prices(p_dairy p_proteins p_fruitveg p_flours p_misc)
> expenditures(expfd) demographics(n_kids n_adults) nolog
. estimates store gquaids
\end{verbatim}

We stored our estimation results for later use by specifying the \texttt{estimates store} command. Next, we obtain each household’s initial utility \( u^0 \) by evaluating the indirect utility function:

\begin{verbatim}
. predict u0, iuf
\end{verbatim}

Now, we create a new dairy price variable containing the price each household will face after the imposition of the tax. Given that price variable and initial utility \( u^0 \), we can calculate each household’s new level of expenditure. Finally, we compute the \( CV \). We type

\begin{verbatim}
. generate ptax_dairy = p_dairy * 1.10
. predict ep1u0, ef utilities(u0)
> prices(ptax_dairy p_proteins p_fruitveg p_flours p_misc)
. generate cv = expfd - ep1u0
\end{verbatim}

Here we used the \texttt{prices()} option of \texttt{predict} to specify the new prices \( p^1 \) at which we want the expenditure function. Had we not specified that option, \texttt{predict} would have used the prices used at estimation, \( p^0 \). Because we wanted to evaluate the function at the level of expenditures \( m \), the same as used at estimation, we did not need to specify the \texttt{expenditures()} option of \texttt{predict}.
We summarize our results by looking at the average CV by household size:

```
.mean cv, over(n_kids)
```

Mean estimation

<table>
<thead>
<tr>
<th>Mean</th>
<th>Std. err.</th>
<th>[95% conf. interval]</th>
<th>Number of obs = 4,160</th>
</tr>
</thead>
<tbody>
<tr>
<td>c.cv@n_kids</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>0</td>
<td>-0.5142129</td>
<td>0.0066562</td>
<td>-0.5272626 to -0.5011631</td>
</tr>
<tr>
<td>1</td>
<td>-0.7762653</td>
<td>0.013486</td>
<td>-0.802705 to -0.7498257</td>
</tr>
<tr>
<td>2</td>
<td>-0.9060989</td>
<td>0.014986</td>
<td>-0.9354795 to -0.8767183</td>
</tr>
<tr>
<td>3</td>
<td>-1.067159</td>
<td>0.0268302</td>
<td>-1.11976 to -1.014557</td>
</tr>
<tr>
<td>4</td>
<td>-1.22238</td>
<td>0.0506472</td>
<td>-1.321675 to -1.123084</td>
</tr>
<tr>
<td>5</td>
<td>-1.550712</td>
<td>0.158197</td>
<td>-1.860863 to -1.240561</td>
</tr>
<tr>
<td>6</td>
<td>-1.304841</td>
<td>0.2223387</td>
<td>-1.740744 to -1.8689384</td>
</tr>
<tr>
<td>7</td>
<td>-1.902748</td>
<td>0.1828158</td>
<td>-2.261165 to -1.544331</td>
</tr>
<tr>
<td>8</td>
<td>-2.923615</td>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td>10</td>
<td>-2.305889</td>
<td>.</td>
<td>.</td>
</tr>
</tbody>
</table>
```

The standard errors of the means are missing for households with 8 and 10 children because only one household of each size exists in our data. For the most part, CV becomes more negative with the number of children, as we expected, because larger households are likely to consume more dairy products by virtue of there being more people within the household. To reiterate, CV is negative here, indicating that the social planner must transfer money to the consumer. For a price decrease, CV would be positive.

**EV** is similar to CV, except that we use the postpolicy level of utility and the initial price vector rather than the initial level of utility and postpolicy price vector. We have

\[
EV = e(p^0, u^1) - e(p^0, u^0) \\
= e(p^0, u^1) - e(p^1, u^1) \\
= e(p^0, u^1) - m
\]

To aid in understanding EV, you may wish to sketch a diagram similar to figure 2. EV tells us how much consumers would be willing to pay to trade at the original price vector rather than the postpolicy one. In other words, EV tells us the change in expenditure that would yield \( u^1 \) if total expenditure were to change rather than the price vector. A negative value indicates that consumers are worse off.

**Example 4**

Continuing the previous example, to calculate EV, we first need to calculate the new level of utility \( u^1 \) each household will enjoy based on the postpolicy dairy price. Because the preferences underlying all the demand systems implemented by demandsys are locally nonsatiated, each household will exhaust its full budget. Thus, the level of expenditure we supply as an argument to the indirect utility function is simply \( m \), the value we used during estimation. To get \( u^1 \), we can therefore call predict, iuf at postpolicy prices.

```
.estimates restore gquaid
(results gquaid are active now)
.predict u1, iuf prices(ptax_dairy p_proteins p_fruitveg p_flours p_misc)
```

We had to restore our estimation results because the means command we used previously overwrites results in e().
Now we evaluate the expenditure function at original prices and our computed level of utility \(u^1\). Because the original prices were used at estimation, we do not need to supply alternative prices via the \(\text{prices()}\) option. We type

\[
. \text{predict ep0u1, ef utilities(u1)}
\]

We are now ready to compute \(EV\) and summarize the results:

\[
. \text{generate ev = ep0u1} - \text{expfd}
. \text{mean ev, over(n_kids)}
\]

<table>
<thead>
<tr>
<th>n_kids</th>
<th>Mean</th>
<th>Std. err.</th>
<th>95% conf. interval</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>-0.5073723</td>
<td>0.0065622</td>
<td>-0.5202378, -0.4945068</td>
</tr>
<tr>
<td>1</td>
<td>-0.7660214</td>
<td>0.013292</td>
<td>-0.7920807, -0.739962</td>
</tr>
<tr>
<td>2</td>
<td>-0.8940226</td>
<td>0.014762</td>
<td>-0.922964, -0.8650811</td>
</tr>
<tr>
<td>3</td>
<td>-1.052524</td>
<td>0.0264206</td>
<td>-1.104322, -1.000725</td>
</tr>
<tr>
<td>4</td>
<td>-1.205508</td>
<td>0.0498826</td>
<td>-1.253304, -1.107711</td>
</tr>
<tr>
<td>5</td>
<td>-1.528722</td>
<td>0.1554489</td>
<td>-1.833485, -1.223959</td>
</tr>
<tr>
<td>6</td>
<td>-1.285955</td>
<td>0.218652</td>
<td>-1.71463, -0.8572807</td>
</tr>
<tr>
<td>7</td>
<td>-1.871663</td>
<td>0.178062</td>
<td>-2.22076, -1.522567</td>
</tr>
<tr>
<td>8</td>
<td>-2.87738</td>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td>9</td>
<td>-1.716629</td>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td>10</td>
<td>.</td>
<td>.</td>
<td>.</td>
</tr>
</tbody>
</table>

For small price changes, \(CV\) and \(EV\) will be similar; however, they will not match because \(CV\) is based on the original level of utility and postpolicy prices and \(EV\) is based on the postpolicy level of utility and original prices. Which one is more appropriate in any situation depends on the policy and goals. If the government is implementing, say, a carbon tax to reduce consumption of goods and services that emit carbon dioxide, \(CV\) may be a better choice because it tells how much the government must compensate consumers so that their overall utilities do not change. If one is evaluating the introduction of a new product competing with an existing product into the market, \(EV\) may be a better metric because it could be interpreted as a measure of how much better off consumers would be thanks to lower prices.

As with our analysis of estimated minimum quantities in example 2, here we have not attempted to obtain measures of the sampling variation associated with the \(CV\) or \(EV\). One alternative to obtain, say, the sampling variation of the mean \(CV\) across households would be to use the bootstrap, being sure to bootstrap the entire procedure, including refitting the model at each bootstrap replication as well as obtaining the desired statistics.

References


Also see

[R] demandsys — Estimation of flexible demand systems

[U] 20 Estimation and postestimation commands